         ;代码清单c19-1
         ;文件名：c19_core.asm
         ;文件说明：保护模式微型核心程序 
         ;创建日期：2022-06-19

         ;以下常量定义部分。内核的大部分内容都应当固定 
         core_code_seg_sel     equ  0x38    ;内核代码段选择子
         core_data_seg_sel     equ  0x30    ;内核数据段选择子 
         sys_routine_seg_sel   equ  0x28    ;系统公共例程代码段的选择子 
         video_ram_seg_sel     equ  0x20    ;视频显示缓冲区的段选择子
         core_stack_seg_sel    equ  0x18    ;内核堆栈段选择子
         mem_0_4_gb_seg_sel    equ  0x08    ;整个0-4GB内存的段的选择子
         idt_linear_address    equ  0x1f000 ;中断描述符表的线性地址
         core_lin_alloc_at     equ  0x80100000
                                            ;内核中可用于分配的起始线性地址
         core_lin_tcb_addr     equ  0x8001f800
                                            ;内核任务TCB的高端线性地址

;-------------------------------------------------------------------------------
         ;以下是系统核心的头部，用于加载核心程序 
         core_length      dd core_end       ;核心程序总长度#00

         sys_routine_seg  dd section.sys_routine.start
                                            ;系统公用例程段位置#04

         core_data_seg    dd section.core_data.start
                                            ;核心数据段位置#08

         core_code_seg    dd section.core_code.start
                                            ;核心代码段位置#0c


         core_entry       dd start          ;核心代码段入口点#10
                          dw core_code_seg_sel

;===============================================================================
         [bits 32]
;===============================================================================
SECTION sys_routine vstart=0                ;系统公共例程代码段 
;-------------------------------------------------------------------------------
         ;字符串显示例程
put_string:                                 ;显示0终止的字符串并移动光标 
                                            ;输入：DS:EBX=串地址
         push ecx

         cli

  .getc:
         mov cl,[ebx]
         or cl,cl
         jz .exit
         call put_char
         inc ebx
         jmp .getc

  .exit:
         sti

         pop ecx
         retf                               ;段间返回

;-------------------------------------------------------------------------------
put_char:                                   ;在当前光标处显示一个字符,并推进
                                            ;光标。仅用于段内调用
                                            ;输入：CL=字符ASCII码
         pushad

         ;以下取当前光标位置
         mov dx,0x3d4
         mov al,0x0e
         out dx,al
         inc dx                             ;0x3d5
         in al,dx                           ;高字
         mov ah,al

         dec dx                             ;0x3d4
         mov al,0x0f
         out dx,al
         inc dx                             ;0x3d5
         in al,dx                           ;低字
         mov bx,ax                          ;BX=代表光标位置的16位数

         cmp cl,0x0d                        ;回车符？
         jnz .put_0a
         mov ax,bx
         mov bl,80
         div bl
         mul bl
         mov bx,ax
         jmp .set_cursor

  .put_0a:
         cmp cl,0x0a                        ;换行符？
         jnz .put_other
         add bx,80
         jmp .roll_screen

  .put_other:                               ;正常显示字符
         push es
         mov eax,video_ram_seg_sel          ;0xb8000段的选择子
         mov es,eax
         shl bx,1
         mov [es:bx],cl
         pop es

         ;以下将光标位置推进一个字符
         shr bx,1
         inc bx

  .roll_screen:
         cmp bx,2000                        ;光标超出屏幕？滚屏
         jl .set_cursor

         push bx                            ;为了修改原书程序的逻辑问题，新增
         push ds
         push es
         mov eax,video_ram_seg_sel
         mov ds,eax
         mov es,eax
         cld
         mov esi,0xa0                       ;小心！32位模式下movsb/w/d
         mov edi,0x00                       ;使用的是esi/edi/ecx
         mov ecx,1920
         rep movsw
         mov bx,3840                        ;清除屏幕最底一行
         mov ecx,80                         ;32位程序应该使用ECX
  .cls:
         mov word[es:bx],0x0720
         add bx,2
         loop .cls

         pop es
         pop ds

         ;mov bx,1920                       ;为了修改原书程序的逻辑问题，删除
         pop bx                             ;为了修改原书程序的逻辑问题，新增
         sub bx,80                          ;为了修改原书程序的逻辑问题，新增

  .set_cursor:
         mov dx,0x3d4
         mov al,0x0e
         out dx,al
         inc dx                             ;0x3d5
         mov al,bh
         out dx,al
         dec dx                             ;0x3d4
         mov al,0x0f
         out dx,al
         inc dx                             ;0x3d5
         mov al,bl
         out dx,al

         popad
         ret

;-------------------------------------------------------------------------------
read_hard_disk_0:                           ;从硬盘读取一个逻辑扇区
                                            ;EAX=逻辑扇区号
                                            ;DS:EBX=目标缓冲区地址
                                            ;返回：EBX=EBX+512
         push eax 
         push ecx
         push edx
      
         push eax

         cli

         mov dx,0x1f2
         mov al,1
         out dx,al                          ;读取的扇区数

         inc dx                             ;0x1f3
         pop eax
         out dx,al                          ;LBA地址7~0

         inc dx                             ;0x1f4
         mov cl,8
         shr eax,cl
         out dx,al                          ;LBA地址15~8

         inc dx                             ;0x1f5
         shr eax,cl
         out dx,al                          ;LBA地址23~16

         inc dx                             ;0x1f6
         shr eax,cl
         or al,0xe0                         ;第一硬盘  LBA地址27~24
         out dx,al

         inc dx                             ;0x1f7
         mov al,0x20                        ;读命令
         out dx,al

  .waits:
         in al,dx
         and al,0x88
         cmp al,0x08
         jnz .waits                         ;不忙，且硬盘已准备好数据传输 

         mov ecx,256                        ;总共要读取的字数
         mov dx,0x1f0
  .readw:
         in ax,dx
         mov [ebx],ax
         add ebx,2
         loop .readw

         sti

         pop edx
         pop ecx
         pop eax
      
         retf                               ;段间返回 

;-------------------------------------------------------------------------------
;汇编语言程序是极难一次成功，而且调试非常困难。这个例程可以提供帮助 
put_hex_dword:                              ;在当前光标处以十六进制形式显示
                                            ;一个双字并推进光标 
                                            ;输入：EDX=要转换并显示的数字
                                            ;输出：无
         pushad
         push ds
      
         mov ax,core_data_seg_sel           ;切换到核心数据段 
         mov ds,ax
      
         mov ebx,bin_hex                    ;指向核心数据段内的转换表
         mov ecx,8
  .xlt:    
         rol edx,4
         mov eax,edx
         and eax,0x0000000f
         xlat
      
         push ecx
         mov cl,al                           
         call put_char
         pop ecx
       
         loop .xlt
      
         pop ds
         popad
         retf

;-------------------------------------------------------------------------------
allocate_a_4k_page:                         ;分配一个4KB的页
                                            ;输入：无
                                            ;输出：EAX=页的物理地址
         push ebx
         push ecx
         push edx
         push ds

         mov eax,core_data_seg_sel
         mov ds,eax

         xor eax,eax
  .b1:
         bts [page_bit_map],eax
         jnc .b2
         inc eax
         cmp eax,page_map_len*8
         jl .b1

         mov ebx,message_3
         call sys_routine_seg_sel:put_string
         hlt                                ;没有可以分配的页，停机

  .b2:
         shl eax,12                         ;乘以4096（0x1000）

         pop ds
         pop edx
         pop ecx
         pop ebx

         ret

;-------------------------------------------------------------------------------
alloc_inst_a_page:                          ;分配一个页，并安装在当前活动的
                                            ;层级分页结构中
                                            ;输入：EBX=页的线性地址
         push eax
         push ebx
         push ecx
         push esi
         push ds

         mov eax,mem_0_4_gb_seg_sel
         mov ds,eax

         ;检查该线性地址所对应的页表是否存在
         mov esi,ebx
         and esi,0xffc00000                 ;清除页表索引和页内偏移部分
         shr esi,20                         ;将页目录索引乘以4作为页内偏移
         or esi,0xfffff000                  ;页目录自身的线性地址+表内偏移

         test dword [esi],0x00000001        ;P位是否为“1”。检查该线性地址是
         jnz .b1                            ;否已经有对应的页表

         ;创建并安装该线性地址所对应的页表
         call allocate_a_4k_page            ;分配一个页做为页表
         or eax,0x00000007
         mov [esi],eax                      ;在页目录中登记该页表

         ;清空当前页表
         mov eax,ebx
         and eax,0xffc00000
         shr eax,10
         or eax,0xffc00000
         mov ecx,1024
  .cls0:
         mov dword [es:eax],0x00000000
         add eax,4
         loop .cls0

  .b1:
         ;检查该线性地址对应的页表项（页）是否存在
         mov esi,ebx
         and esi,0xfffff000                 ;清除页内偏移部分
         shr esi,10                         ;将页目录索引变成页表索引，页表索引乘以4作为页内偏移
         or esi,0xffc00000                  ;得到该线性地址对应的页表项

         test dword [esi],0x00000001        ;P位是否为“1”。检查该线性地址是
         jnz .b2                            ;否已经有对应的页

         ;创建并安装该线性地址所对应的页
         call allocate_a_4k_page            ;分配一个页，这才是要安装的页
         or eax,0x00000007
         mov [esi],eax

  .b2:
         pop ds
         pop esi
         pop ecx
         pop ebx
         pop eax

         retf

;-------------------------------------------------------------------------------
create_copy_cur_pdir:                       ;创建新页目录，并复制当前页目录内容
                                            ;输入：无
                                            ;输出：EAX=新页目录的物理地址
         push ds
         push es
         push esi
         push edi
         push ebx
         push ecx

         mov ebx,mem_0_4_gb_seg_sel
         mov ds,ebx
         mov es,ebx

         call allocate_a_4k_page
         mov ebx,eax
         or ebx,0x00000007
         mov [0xfffffff8],ebx

         invlpg [0xfffffff8]

         mov esi,0xfffff000                 ;ESI->当前页目录的线性地址
         mov edi,0xffffe000                 ;EDI->新页目录的线性地址
         mov ecx,1024                       ;ECX=要复制的目录项数
         cld
         repe movsd

         pop ecx
         pop ebx
         pop edi
         pop esi
         pop es
         pop ds

         retf

;-------------------------------------------------------------------------------
task_alloc_memory:                          ;在指定任务的虚拟内存空间中分配内存
                                            ;输入：EBX=任务控制块TCB的线性地址
                                            ;      ECX=希望分配的字节数
                                            ;输出：ECX=已分配的起始线性地址
         push eax

         push ds

         push ebx                           ;to A

         ;获得本次内存分配的起始线性地址
         mov ax,mem_0_4_gb_seg_sel
         mov ds,ax

         mov ebx,[ebx+0x46]                 ;获得本次分配的起始线性地址
         mov eax,ebx
         add ecx,ebx                        ;本次分配，最后一个字节之后的线性地址

         push ecx                           ;to B

         ;为请求的内存分配页
         and ebx,0xfffff000
         and ecx,0xfffff000
  .next:
         call sys_routine_seg_sel:alloc_inst_a_page
                                            ;安装当前线性地址所在的页
         add ebx,0x1000                     ;+4096
         cmp ebx,ecx
         jle .next

         ;将用于下一次分配的线性地址强制按4字节对齐
         pop ecx                            ;B

         test ecx,0x00000003                ;线性地址是4字节对齐的吗？
         jz .algn                           ;是，直接返回
         add ecx,4                          ;否，强制按4字节对齐
         and ecx,0xfffffffc

  .algn:
         pop ebx                            ;A

         mov [ebx+0x46],ecx                 ;将下次分配可用的线性地址回存到TCB中
         mov ecx,eax

         pop ds

         pop eax

         retf

;-------------------------------------------------------------------------------
allocate_memory:                            ;在当前任务的地址空间中分配内存
                                            ;输入：ECX=希望分配的字节数
                                            ;输出：ECX=起始线性地址 
         push eax
         push ebx

         push ds

         ;得到TCB链表首节点的线性地址
         mov eax,core_data_seg_sel
         mov ds,eax

         mov eax,[tcb_chain]                ;EAX=首节点的线性地址

         mov ebx,mem_0_4_gb_seg_sel
         mov ds,ebx

         ;搜索状态为忙（当前任务）的节点
  .s0:
         cmp word [eax+0x04],0xffff
         jz .s1                             ;找到忙的节点，EAX=节点的线性地址
         mov eax,[eax]
         jmp .s0

         ;开始分配内存
  .s1:
         mov ebx,eax
         call sys_routine_seg_sel:task_alloc_memory

         pop ds

         pop ebx
         pop eax

         retf

;-------------------------------------------------------------------------------
set_up_gdt_descriptor:                      ;在GDT内安装一个新的描述符
                                            ;输入：EDX:EAX=描述符 
                                            ;输出：CX=描述符的选择子
         push eax
         push ebx
         push edx

         push ds
         push es

         mov ebx,core_data_seg_sel          ;切换到核心数据段
         mov ds,ebx

         sgdt [pgdt]                        ;以便开始处理GDT

         mov ebx,mem_0_4_gb_seg_sel
         mov es,ebx

         movzx ebx,word [pgdt]              ;GDT界限
         inc bx                             ;GDT总字节数，也是下一个描述符偏移
         add ebx,[pgdt+2]                   ;下一个描述符的线性地址

         mov [es:ebx],eax
         mov [es:ebx+4],edx

         add word [pgdt],8                  ;增加一个描述符的大小

         lgdt [pgdt]                        ;对GDT的更改生效

         mov ax,[pgdt]                      ;得到GDT界限值
         xor dx,dx
         mov bx,8
         div bx                             ;除以8，去掉余数
         mov cx,ax
         shl cx,3                           ;将索引号移到正确位置

         pop es
         pop ds

         pop edx
         pop ebx
         pop eax

         retf
;-------------------------------------------------------------------------------
make_seg_descriptor:                        ;构造存储器和系统的段描述符
                                            ;输入：EAX=线性基地址
                                            ;      EBX=段界限
                                            ;      ECX=属性。各属性位都在原始
                                            ;          位置，无关的位清零 
                                            ;返回：EDX:EAX=描述符
         mov edx,eax
         shl eax,16
         or ax,bx                           ;描述符前32位(EAX)构造完毕

         and edx,0xffff0000                 ;清除基地址中无关的位
         rol edx,8
         bswap edx                          ;装配基址的31~24和23~16  (80486+)

         xor bx,bx
         or edx,ebx                         ;装配段界限的高4位

         or edx,ecx                         ;装配属性

         retf

;-------------------------------------------------------------------------------
make_gate_descriptor:                       ;构造门的描述符（调用门等）
                                            ;输入：EAX=门代码在段内偏移地址
                                            ;       BX=门代码所在段的选择子 
                                            ;       CX=段类型及属性等（各属
                                            ;          性位都在原始位置）
                                            ;返回：EDX:EAX=完整的描述符
         push ebx
         push ecx
      
         mov edx,eax
         and edx,0xffff0000                 ;得到偏移地址高16位 
         or dx,cx                           ;组装属性部分到EDX
       
         and eax,0x0000ffff                 ;得到偏移地址低16位 
         shl ebx,16                          
         or eax,ebx                         ;组装段选择子部分
      
         pop ecx
         pop ebx
      
         retf                                   
                             
;-------------------------------------------------------------------------------
initiate_task_switch:                       ;主动发起任务切换
                                            ;输入：无
                                            ;输出：无。
         pushad
         push ds
         push es

         mov eax,core_data_seg_sel
         mov es,eax

         mov eax,mem_0_4_gb_seg_sel
         mov ds,eax

         mov eax,[es:tcb_chain]
         cmp eax,0
         jz .return

         ;搜索状态为忙（当前任务）的节点
  .b0:
         cmp word [eax+0x04],0xffff
         cmove esi,eax                     ;找到忙的节点，ESI=节点的线性地址
         jz .b1
         mov eax,[eax]
         jmp .b0

         ;从当前节点继续搜索就绪任务的节点
  .b1:
         mov ebx,[eax]
         or ebx,ebx
         jz .b2                            ;到链表尾部也未发现就绪节点，从头找
         cmp word [ebx+0x04],0x0000
         cmove edi,ebx                     ;已找到就绪节点，EDI=节点的线性地址
         jz .b3
         mov eax,ebx
         jmp .b1

  .b2:
         mov ebx,[es:tcb_chain]            ;EBX=链表首节点线性地址
  .b20:
         cmp word [ebx+0x04],0x0000
         cmove edi,ebx                     ;已找到就绪节点，EDI=节点的线性地址
         jz .b3
         mov ebx,[ebx]
         or ebx,ebx
         jz .return                        ;链表中已经不存在就绪任务，返回
         jmp .b20

         ;就绪任务的节点已经找到，准备切换到该任务
  .b3:
         not word [esi+0x04]               ;将忙状态的节点改为就绪状态的节点
         not word [edi+0x04]               ;将就绪状态的节点改为忙状态的节点
         jmp far [edi+0x14]                ;任务切换

  .return:
         pop es
         pop ds
         popad

         retf

;-------------------------------------------------------------------------------
terminate_current_task:                     ;终止当前任务
                                            ;注意，执行此例程时，当前任务仍在
                                            ;运行中。此例程其实也是当前任务的
                                            ;一部分 
         mov eax,core_data_seg_sel
         mov es,eax

         mov eax,mem_0_4_gb_seg_sel
         mov ds,eax

         mov eax,[es:tcb_chain]
                                            ;EAX=首节点的线性地址
         ;搜索状态为忙（当前任务）的节点
  .s0:
         cmp word [eax+0x04],0xffff
         jz .s1                             ;找到忙的节点，EAX=节点的线性地址
         mov eax,[eax]
         jmp .s0

         ;将状态为忙的节点改成终止状态
  .s1:
         mov word [eax+0x04],0x3333

         ;搜索就绪状态的任务
         mov ebx,[es:tcb_chain]            ;EBX=链表首节点线性地址
  .s2:
         cmp word [ebx+0x04],0x0000
         jz .s3                            ;已找到就绪节点，EBX=节点的线性地址
         mov ebx,[ebx]
         jmp .s2

         ;就绪任务的节点已经找到，准备切换到该任务
  .s3:
         not word [ebx+0x04]               ;将就绪状态的节点改为忙状态的节点
         jmp far [ebx+0x14]                ;任务切换

;-------------------------------------------------------------------------------
general_interrupt_handler:                  ;通用的中断处理过程
         push eax

         mov al,0x20                        ;中断结束命令EOI
         out 0xa0,al                        ;向从片发送
         out 0x20,al                        ;向主片发送

         pop eax

         iretd

;-------------------------------------------------------------------------------
general_exception_handler:                  ;通用的异常处理过程
         mov ebx,excep_msg
         call sys_routine_seg_sel:put_string

         cli

         hlt

;-------------------------------------------------------------------------------
rtm_0x70_interrupt_handle:                  ;实时时钟中断处理过程

         pushad

         mov al,0x20                        ;中断结束命令EOI
         out 0xa0,al                        ;向8259A从片发送
         out 0x20,al                        ;向8259A主片发送

         mov al,0x0c                        ;寄存器C的索引。且开放NMI
         out 0x70,al
         in al,0x71                         ;读一下RTC的寄存器C，否则只发生一次中断
                                            ;此处不考虑闹钟和周期性中断的情况
         ;请求任务调度
         call sys_routine_seg_sel:initiate_task_switch

         popad

         iretd

;-------------------------------------------------------------------------------
do_task_clean:                             ;清理已经终止的任务并回收资源

         ;搜索TCB链表，找到状态为终止的节点
         ;将节点从链表中拆除
         ;回收任务占用的各种资源（可以从它的TCB中找到）

         retf

sys_routine_end:

;===============================================================================
SECTION core_data vstart=0                  ;系统核心的数据段 
;------------------------------------------------------------------------------- 
         pgdt             dw  0             ;用于设置和修改GDT 
                          dd  0

         pidt             dw  0
                          dd  0

         page_bit_map     db  0xff,0xff,0xff,0xff,0xff,0xff,0x55,0x55
                          db  0xff,0xff,0xff,0xff,0xff,0xff,0xff,0xff
                          db  0xff,0xff,0xff,0xff,0xff,0xff,0xff,0xff
                          db  0xff,0xff,0xff,0xff,0xff,0xff,0xff,0xff
                          db  0x55,0x55,0x55,0x55,0x55,0x55,0x55,0x55
                          db  0x00,0x00,0x00,0x00,0x00,0x00,0x00,0x00
                          db  0x00,0x00,0x00,0x00,0x00,0x00,0x00,0x00
                          db  0x00,0x00,0x00,0x00,0x00,0x00,0x00,0x00
         page_map_len     equ $-page_bit_map

         ;符号地址检索表
         salt:
         salt_1           db  '@PrintString'
                     times 256-($-salt_1) db 0
                          dd  put_string
                          dw  sys_routine_seg_sel

         salt_2           db  '@ReadDiskData'
                     times 256-($-salt_2) db 0
                          dd  read_hard_disk_0
                          dw  sys_routine_seg_sel

         salt_3           db  '@PrintDwordAsHexString'
                     times 256-($-salt_3) db 0
                          dd  put_hex_dword
                          dw  sys_routine_seg_sel

         salt_4           db  '@TerminateProgram'
                     times 256-($-salt_4) db 0
                          dd  terminate_current_task
                          dw  sys_routine_seg_sel

         salt_5           db  '@InitTaskSwitch'
                     times 256-($-salt_5) db 0
                          dd  initiate_task_switch
                          dw  sys_routine_seg_sel

         salt_6           db  '@malloc'
                     times 256-($-salt_6) db 0
                          dd  allocate_memory
                          dw  sys_routine_seg_sel

         salt_item_len   equ $-salt_6
         salt_items      equ ($-salt)/salt_item_len

         message_0        db  '  System core is runing in protect mode,'
                          db  'IDT is mounted.',0x0d,0x0a,0

         cpu_brnd0        db  0x0d,0x0a,'  ',0
         cpu_brand  times 52  db 0
         cpu_brnd1        db  0x0d,0x0a,0x0d,0x0a,0

         message_1        db  '  Paging is enabled.System core is mapped to'
                          db  ' linear address 0x80000000.',0x0d,0x0a,0

         message_2        db  '  System wide CALL-GATE mounted and test OK.'
                          db  0x0d,0x0a,0

         message_3        db  '********No more pages********',0

         excep_msg        db  '********Exception encounted********',0

         bin_hex          db '0123456789ABCDEF'
                                            ;put_hex_dword子过程用的查找表 

         core_buf   times 2048 db 0         ;内核用的缓冲区

         ;任务控制块链
         tcb_chain        dd  0

         core_msg1        db  '  Core task created.',0x0d,0x0a,0
                 
         core_msg2        db  '[CORE TASK]: I am working!',0x0d,0x0a,0

core_data_end:
               
;===============================================================================
SECTION core_code vstart=0
;-------------------------------------------------------------------------------
fill_descriptor_in_ldt:                     ;在LDT内安装一个新的描述符
                                            ;输入：EDX:EAX=描述符
                                            ;          EBX=TCB基地址
                    